Introduction
It is widely recognized that fixed-effects models have an advantage over random-effects models when analyzing panel data because they control for all level 2 characteristics, measured or unmeasured (Allison 2009; Halaby 2004; Wooldridge 2010) . This also applies in a multilevel framework. However, a major drawback of fixed-effects models is their inability to estimate the effect of any variable that does not vary within clusters, which holds for all level 2 variables. To circumvent this disadvantage, it has been proposed to estimate within effects in random-effects models (Allison 2009; Neuhaus and Kalbfleisch 1998; Rabe-Hesketh and Skrondal 2008; Raudenbush 1989a; Wooldridge 2010 ).
Models
In the linear case, 1 the random-intercept model is given by
where subscript i denotes level 2 (for example, subjects) and t denotes level 1 (for example, occasions). x it is a level 1 variable that varies between and within clusters, and c i is a level 2 variable that varies only between clusters. µ i is the level 2 error and the random intercept, and ǫ it is the level 1 error. Throughout the article, ǫ it will be treated as white noise and not considered further.
The standard distributional assumption regarding the level 2 error is µ i |x it , c i ∼ N (0, σ 2 µ ). The model provides consistent effect estimates if E(µ i |x it , c i ) = 0. Subtracting the between model
from (1) provides the fixed-effects model in its demeaned form:
The subtraction removes the level 2 error (µ i ) from the equation. As a result, the model's estimate of β 1 is unbiased even if E(µ i |x it ) = 0. But this comes at a cost.
The subtraction also removes all variables that do not vary at level 1. Fixed-effects models therefore cannot estimate the effect of level 2 variables. This may not be seen as a problem in panel-data analysis. But it is definitely a major drawback in multilevel analysis, where interest often lies particularly in estimating the effect of level 2 variables.
However, it is possible to estimate within effects in random-effects models by decomposing level 1 variables into a between (x i = n −1 i ni t=1 x it ) and a cluster (x it − x i ) component. This hybrid model (Allison 2009 ) is given by
Using (3) to estimate β 1 gives the within-effect estimate, that is, the fixed-effects estimate (Mundlak 1978; Neuhaus and Kalbfleisch 1998) . Hence, the estimates of β 1 from (2) and (3) are identical. Because (3) is a random-effects model, we can use it to estimate effects of level 2 variables. However, for the estimate of β 2 to be unbiased, E(µ i |x it , c i ) = 0 and µ i |x it , c i ∼ N (0, σ 2 µ ) still have to hold. Moreover, even though (3) is a random-effects model, its estimate of β 1 is not more efficient than the one obtained through estimating (2), because both estimates are solely based on within variation. β 3 estimates the between effect (Mundlak 1978; Neuhaus and Kalbfleisch 1998) . While it is not necessary to include the cluster mean (x i ) to obtain the within estimate of β 1 , its inclusion ensures that effect estimates of level 2 variables are corrected for between-cluster differences in x it .
The idea to decompose between and within variation and to estimate the respective effects in a single model is not new (Kaufman 1993; Kreft, de Leeuw, and Aiken 1995; Neuhaus and Kalbfleisch 1998; Raudenbush 1989a ), but it seems to have become increasingly popular in panel data (Burnett and Farkas 2009; Phillips 2006; Ousey and Wilcox 2007; Teachman 2011; Zhou 2011) as well as in multilevel analysis (Curran and Bauer 2011; Epstein et al. 2012; Landale, Gorman, and Oropesa 2006; Nomaguchi and Brown 2011; Park, Lee, and Epstein 2009; Schempf et al. 2011 ).
This approach offers several additional advantages. First, it allows us to test for equivalence of within and between estimates. This test, which is referred to as an augmented regression test (Jones et al. 2007, 217) , can be used as an alternative to the Hausman specification test (Baltagi 2008, 73) . If between and within effects are the same-that is, β 1 = β 3 -then (3) collapses to (1), the random-intercept model. Second, a decomposition into between and within effects can be used with generalized estimating equations, which enables us to specify less restrictive within-cluster error structures. Third, this approach can be extended to include random slopes, allowing effects of level 1 variables to vary between clusters.
A hybrid model that includes a random slope for (x it − x i ) is given by
The hybrid model is closely related to the correlated random-effects model (Wooldridge 2010) , first proposed by Mundlak (1978) . The correlated random-effects model relaxes the assumption of zero correlation between the level 2 error and the level 1 variables. In particular, it introduces the assumption µ i = πx i + ν i , so (1) becomes
The cluster mean of x it picks up any correlation between this variable and the level 2 error. Including the cluster mean of a level 1 variable in a random-effects model is therefore an alternative to cluster mean centering (Halaby 2003, 519) . Thus β 1 from (4) is the fixed-effects estimate (Mundlak 1978; Wooldridge 2010) , and it is identical to the estimate obtained from (3). But the estimated effect of x i will differ. In the hybrid model, this is the between effect. In the correlated random-effects model, this is the difference of the within and between effects (Mundlak 1978) ; that is, π = β 3 − β 1 .
The relation between the hybrid model and the correlated random-effects model becomes obvious if we rewrite (3) as
The correlated random-effects model allows for the inclusion of level 2 variables, and it can be used with generalized estimating equations, just like the hybrid model. It is also possible to perform an augmented regression test. But the test takes a different form. Because π already estimates the difference of the within and between effects, the null hypothesis is π = 0 (Baltagi 2008, 133) . In principle, the correlated random-effects model can also include random slopes. However, a correlated random-effects model with random slopes is not equivalent to the corresponding hybrid model with random slopes, and it is advisable to use the hybrid model (Raudenbush 1989a; 1989b) . 2. In particular, because π = β 3 − β 1 , the random part of the slope, µ 2i , will appear in the estimated effect of x it and x i , which makes it hard to interpret (see Raudenbush [1989a, 12] ).
Estimation
The example presented below uses infant birth weight data (Abrevaya 2006; RabeHesketh and Skrondal 2008) . The data comprise 8,604 infants clustered within 3,978 mothers. We will examine how the level 1 variables, mother's age (mage, continuous) and smoking behavior (smoke, binary), and the level 2 variable, race (black, binary), affect a child's birth weight (birwt, continuous). To estimate between and within effects in one model, we must first generate the cluster-specific mean of x it . The second step is to create the deviation scores, which is also known as group mean centering. We have to ensure that the means are generated on the multivariate sample, that is, by using listwise deletion to handle missings. This is done using mark (Jann 2007b).
. use http://www.stata-press.com/data/mlmus2/smoking . mark nonmiss . markout nonmiss birwt smoke mage black . egen msmoke = mean(smoke) if nonmiss==1, by(momid) . generate dsmoke = smoke-msmoke
The cluster-specific means and the deviation scores can also be computed easily with the center command (Jann 2007a) .
.
by momid, sort: center mage if nonmiss==1, prefix(d) mean(m)
Once the variables have been created, xtreg, re can be used to fit the hybrid model.
. xtreg birwt dsmoke dmage msmoke mmage black, i(momid) re . estimates store hybrid
The correlated random-effects model is fit in a similar way, but it includes uncentered versions of the level 1 variables.
. xtreg birwt smoke mage msmoke mmage black, i(momid) re . estimates store corr_re
We also fit a random-intercept and a fixed-effects model so that we can compare their estimates with those of the hybrid model and the correlated random-effects model.
. xtreg birwt smoke mage black, i(momid) re . estimates store re . xtreg birwt smoke mage, i(momid) fe . estimates store fe Estimation results are shown in table 1. Model 1 presents the estimates from the random-intercept model, model 2 from the fixed-effects model, model 3 from the hybrid model, and model 4 from the correlated random-effects model. Comparing the randomintercept model (model 1) with the fixed-effects model (model 2), we see that there are considerable differences in effect estimates. The estimated effect on birth weight of smoking during pregnancy, for instance, is considerably smaller when estimated as a within-mother effect, that is, once all time-constant (observed and unobserved) differences between mothers are accounted for. The estimated effect of mother's age, on the other hand, is considerably larger in the fixed-effects model than in the random-effects model.
If we compare the estimates from the fixed-effects model (model 2) with those from the hybrid model (model 3) and the correlated random-effects model (model 4), we see that all three models estimate the same (within) effects of the level 1 variables.
3 A comparison of between and within effects from the hybrid model (model 3) provides additional insight. The model, for instance, estimates the between-mother effect of smoking as −332.93. Accordingly, the average birth weight for a mother who smokes and a mother who does not smoke will differ by 332.928 grams. The within-mother effect is estimated to be −105.70. Thus for a given mother, smoking decreases the birth weight of her children by 105.70 on average. A Wald test can be used to test for equivalence of within and between estimates.
. estimates restore hybrid . test dsmoke=msmoke . test dmage=mmage . hausman fe re
In the present case, the test statistics suggest that the null hypothesis of equality for within and between estimates should be rejected [smoking: Wald χ 2 (1) = 38.68, age: Wald χ 2 (1) = 22.09], which can be considered evidence against the random-effects model. A Hausman test reaches the same conclusion [χ 2 (2) = 58.63]. How can we explain the differences of between and within effects? Smoking is likely to be correlated with other mother-specific unobserved variables (Abrevaya 2006 ) that adversely impact birth weight. Therefore, the between effect (and the estimate from the randomintercept model, which is a weighted average of the between and the within estimate) overestimates the effect of smoking. Certainly, this raises the question whether there is a meaningful interpretation of the between effect. The estimate is obviously biased because it is confounded with the level 2 error. However, a comparison with the within estimate can inform us how much of the observed relation in birth weight and a mother's smoking is due to unobserved heterogeneity between smoking and nonsmoking mothers, which is not accounted for by our model. As argued above, the estimated effects of the cluster means will differ between the hybrid model and the correlated random-effects model. In the correlated random-effects model, this is π = β 3 − β 1 , as shown in (5). Indeed, the estimated effect of −227.23 for the cluster mean from the correlated random-effects model (model 4) corresponds exactly to the difference of the estimated within-mother and between-mother effects from the hybrid model (model 3) As columns 4 and 5 in table 1 show, the hybrid model and the correlated randomeffects model also provide (identical) effect estimates of the level 2 variable black. The estimated effect of this variable is similar, albeit not identical, to the one obtained from the random-intercept model (model 1). This is because including x i controls more encompassingly for between-cluster differences in x it .
As pointed out above, a decomposition into between and within effects also allows us to incorporate random slopes. Let us assume we have reasons to believe that the within effect of smoking varies across mothers. In that case, we would want to specify a hybrid model with a random slope for the variable dsmoke. This can be done with xtmixed.
. xtmixed birwt dsmoke dmage msmoke mmage black || momid: dsmoke
The results are shown in the last column of table 1 (model 5).
Interactions
There are pitfalls to the application of these models when including interactions. Let us say we are interested in including an interaction of smoking by mother's age. Thus our comparison model is the following fixed-effects model:
We can fit this easily by using the operator # to specify the interaction, the c. operator to indicate continuous variables, and i. to indicate factor variables.
. xtreg birwt i.smoke##c.mage, i(momid) fe . estimates store fe_inter
However, specifying the hybrid model as
4
. xtreg birwt c.dsmoke##c.dmage c.msmoke##c.mmage black, i(momid) re . estimates store hybrid_inter_incorrect is incorrect. Why is that? What we want to estimate is β k (x it z it − x i z i ). But if we specify the interaction in the hybrid model as above, we estimate
, which produces a completely different result. Therefore, we first have to generate the interaction term x it z it , cluster mean center the new variable, and then enter it into the model.
. generate smokeXmage = smoke*mage . by momid, sort: center smokeXmage if nonmiss==1, prefix(d) mean(m) . xtreg birwt dsmoke dmage dsmokeXmage msmoke mmage msmokeXmage black, i(momid) re . estimates store hybrid_inter_correct
Because the correlated random-effects model does not include deviations from the cluster means but does include the variables in their uncentered form, it is possible to include the interaction of these variables with the operator #. However, to obtain the correct within effect of the interaction, we still have to include the respective interaction terms of the cluster means. This, again, cannot be done with the operator #. We have to control for x i z i , but using # instead results in x i z i .
The following specification therefore results in incorrect estimates:
. xtreg birwt i.smoke##c.mage c.msmoke##c.mmage black, i(momid) re . estimates store corr_re_inter_incorrect
The correct estimates are obtained through the following specification:
. xtreg birwt i.smoke##c.mage msmoke mmage msmokeXmage black, i(momid) re . estimates store corr_re_inter_correct An estimation example is provided in table 2. Model 1 shows the estimates from the standard fixed-effects model. This model estimates the interaction effect of smoking by mother's age as 3.79. Models 2 and 3 present estimates from hybrid models, and models 4 and 5 present estimates from correlated random-effects models. Models 2 and 4 include the incorrect interaction, and models 3 and 5 include the correct interactions. If we compare the estimated interaction effects of models 2 and 4 with the benchmark, model 1, we see that both are incorrect. Moreover, the estimated main effects of models 2 and 4 are, of course, also incorrect. Models 3 and 5, on the other hand, estimate the correct main and interaction effects. 4. Note that group mean-centered variables are continuous. 5. The magnitude of the difference between the correct and the incorrect estimates is considerably larger in the hybrid model than in the correlated random-effects model. This is what we would expect, considering that the difference between the correct and the incorrect interaction terms in the hybrid model, that is, the difference between (x it z it − x i z i ) and {(x it − x i )(z it − z i )}, is larger than the difference in the correlated random-effects model, that is, the difference between x i z i and x i z i . If we want to exploit the possibility of estimating within effects in random-effects models via the hybrid specification, we have to include interaction terms in the oldfashioned way by generating interaction variables. If we use the correlated randomeffects setup, we still have to generate interactions of the cluster mean variables by hand.
Unfortunately, in the case of the hybrid model, it becomes impossible to use postestimation commands such as margins in the usual manner. margins relies on factorvariable notation. If interactions are not specified via #, Stata will not automatically take into account both main and interaction effects.
Suppose we are interested in the marginal effect of mother's age, that is, in the partial derivate of (6) with respect to mother's age. Based on the fixed-effects model (model 1) this is 23.28. However, using margins after the hybrid model (model 3) gives 22.75, which is only the main effect. Using margins after the correlated random-effects model (model 5), on the other hand, gives the correct estimate of 23.28.
6
. estimates restore fe_inter . margins, dydx(mage) . estimates restore hybrid_inter_correct . margins, dydx(dmage) . estimates restore corr_re_inter_correct . margins, dydx(mage)
Conclusion
I discussed some advantages that correlated random-effects and hybrid models offer. In either case, a decomposition of within and between effects in a single model increases flexibility in model setup because it combines advantages of fixed-and random-effects models. It allows us to estimate the effect of level 2 variables while providing effect estimates of level 1 variables that are unbiased by a possible correlation with the level 2 error. Moreover, a comparison of within and between effects (or their difference) provides an assessment of the degree to which unobserved heterogeneity in level 2 characteristics is responsible for an observable relation between the outcome and a level 1 variable, which is not accounted for by our model. Importantly, these advantages apply when handling any clustered data, whether it is panel data or multilevel data.
Yet there are some aspects to consider. First, within-effect estimates obtained through random-effects models are not more efficient than those obtained from fixedeffects models. Second, the approach described here offers no remedy for a possible correlation of a level 2 variable and the level 2 error (this would require an instrumentalvariables approach; see Hausman and Taylor [1981] when handling panel data). Third, handling interactions in these models may be cumbersome. Nevertheless, these models are useful extensions to the standard random-effects and fixed-effects approaches.
